DESIGNED FOR HP-IB SYSTEMS

Programming Note

8350A/85-1
Supersedes: None

Introductory Operating Guide for the HP 8350A Sweep Oscillator with the HP 85A Personal Computer



INTRODUCTION

This programming note is a guide to the remote operation of the HP 8350A Sweep Oscillator and appropriate HP 83500 Series Plug-in using the HP 85A Personal Computer. Included in this guide are the system connections for remote operation and several example programs with descriptions of each step.

The 8350A is fully compatible with the Hewlett-Packard Interface Bus (HP-IB). When used with a controller such as the 85A, complete control of the sweep mode, frequency limits, frequency markers, power level, and all other front panel controls can be achieved.

REFERENCE INFORMATION

For further information on the HP Interface Bus, the following references should prove helpful:

- Condensed Description of the Hewlett-Packard Interface Bus (HP Literature No. 59401-90030).

Complete reference information on the 8350A can be found in the 8350A Sweep Oscillator Operating and Service Manual (HP Part No. 08350-90001). For information on operating the 85A the following references are available:

- 85A Owner's Manual and Programming Guide (HP Part No. 00085-90002).
- 85A I/O Programming Guide (HP Part No. 00085-90142).

HEWLETT PACKARD

EQUIPMENT REQUIRED

To perform all the example programs as described in this programming note, you will need the following equipment and accessories:

1. HP 8350A Sweep Oscillator with any HP 83500 Series Plug-in. Note that an HP 86200 Series Plug-in with the HP 11869A Adapter can be used but all references to power level and power control are not applicable.

2. HP 85A Personal Computer with:

   a. HP Part No. 00085-15003 I/O ROM

   b. HP 82936A ROM Drawer

   c. HP 82937A HP-IB Interface Card/Cable

NOTE

The following equipment is not required for the programs to function but rather for a visual display of the 8350A functions.

3. HP 8755S Frequency Response Test Set with:

   a. HP 8755C Swept Amplitude Analyzer

   b. HP 180TR or 182T Display Unit

   c. HP 11664A or 11664B Detector

   d. Two 120 centimetre BNC cables (HP 11170C variety)

   or any appropriate Oscilloscope with Crystal/ Schottky Detector, Attenuator, and BNC Cabling.

4. Any test device over the frequency range of the 83500 Series Plug-in.



FIGURE 1: System Connection

SET-UP

Figure 1 shows the system connection and switch settings for the 82937A HP–IB Interface and the 85A Personal Computer. The following procedure completes the setup:

1. Turn off the power to the 85A.
2. Verify that the ROM is installed in the 85A. If not, then install the ROM in the 82936A ROM Drawer then insert the drawer in one of the rear panel slots of the 85A.
3. Install the 82937A HP–IB Interface Card into one of the rear panel slots of the 85A.
4. Connect the 24-pin HP-IB connector of the 82937A to the rear panel HP-IB connector of the 8350A. This connector is tapered to insure proper connection.

   CAUTION

   Do not attempt to mate black metric threaded screws on one connector with silver English threaded nuts on another connector, or vice-versa, as damage may result. A metric conversion kit which will convert one cable and one or two instruments to metric hardware is available by ordering HP Part No. 5060-0138.

5. All programs within this guide expect the 8350A HP-IB address to be decimal 19. The 8350A HP-IB address switches are located inside the instrument and are factory preset to decimal 19. To find the present HP-IB address use the front panel "Set HP-IB Address" by executing:

   Press SHIFT LCL

   The FREQUENCY/TIME display will indicate the present decimal address. To reset the number displayed if not 19:

   Press 1 9 GHz

   This HP-IB address will remain in effect until the instrument is powered off since the internal address switches are read at power on (unless 8350A Option 001 Non-volatile Memory is used). Since Example 4 requires the 8350A to be powered off and then on, the internal address switches should be reset to 19 if necessary.

CHECK-OUT

Turn on the 85A and the 8350A. The 85A should display the cursor ("–") in the upper left corner of the CRT display. The 8350A should undergo a turn-on self test consisting of the red LED numeric displays being blanked and all yellow indicator LED's on, then the 8350A sweep controls are set to the instrument preset state: Start/Stop Sweep over the entire plug-in frequency range, fastest sweep time for plug-in (typically 10 milliseconds), and maximum leveled output power for the plug-in. If the 8350A fails the power-on self test an error message will be displayed in the far left LED display. Check section 8 of the 8350A Operating and Service Manual for error message decoding and diagnostics.

To verify that the HP-IB connections and interface are functional perform the following on the 85A:

1. Press SHIFT RESET
2. Type 'REMOTE 719'
3. Press END LINE

Verify that the REMote light on the 8350A is lit. If this fails, verify that the 82937A select code switch is set to "7" (this switch is located inside the 82937A so refer to its Installation Manual), the 8350A address switches are set to "19", and the interface cable is properly connected.

If the 85A display indicates an error message, it is possible that the above remote message was typed in incorrectly or the ROM's are not properly installed. If the 85A accepts the remote statement and the display is clear but the 8350A REMote light does not turn on, you could have a defective 82937A or 8350A. Perform the operational checks as outlined in the respective Operating and Service Manuals to find the defective device.

PROGRAMMING EXAMPLES

The following sample programs show the various ways of controlling the 8350A. In remote control situations the 8350A Sweep Oscillator can interact with the system HP-IB controller in two basic ways:

1. "Listen Mode": The 8350A listens to the control commands as to modifying the present instrument state. This effectively commands the 8350A to do a specific event much like setting a front panel function.
2. "Talk Mode": The 8350A informs the controller of the present instrument state with a numeric value or a string of characters. This effectively allows the user to interrogate or learn any 8350A function.

Each programming example is structured using the following format:

1. A general description of the functions exercised.
2. The program listing.
3. An explanation of each program line.
4. Detailed instructions for operating the system.

EXAMPLE PROGRAM 1: Remote, Local, Local Lockout, and Instrument Preset

Before programming the 8350A for different sweep functions, the user should be aware of the extent of remote control that can be used. The Remote Enable ('REMOTE') command sets the 8350A into remote control from the local (manual) mode. In remote the 8350A will perform only as its functions are programmed. However if the LOCAL button is pressed, the 8350A will return from the remote state to local control. To prevent this from occurring the Local Lockout ('LOCAL LOCKOUT') command disables all front panel controls, specifically the "Local" key. The Go To Local ('LOCAL') command will return the 8350A to front panel control thereby removing it from the remote and local lockout modes. Note that the above remote and local commands are different from the general HP-IB bus local and remote commands ('LOCAL 7' and 'REMOTE 7'). Finally, in remote control it is periodically desirable to reset the 8350A to a predefined state, this is achievable with the Instrument Preset function.

PROGRAM 1

```
10 REMOTE 719
20 DISP "Remote"
30 PAUSE
40 REMOTE 719
50 LOCAL LOCKOUT 7
60 DISP "Local Lockout"
70 PAUSE
80 LOCAL 719
90 DISP "Local"
100 PAUSE
110 OUTPUT 719 ;"IP"
120 END
```

PROGRAM 1 EXPLANATION:

Line 10: Sets 8350A to remote.
Line 20: The 85A displays "Remote'.
Line 30: Temporarily stops program execution.
Line 40: Sets 8350A to remote.
Line 50: Sets local lockout mode.
Line 60: The 85A displays "Local Lockout".
Line 70: Temporarily stops program execution.
Line 80: Sets 8350A to local.
Line 90: The 85A displays "Local".
Line 100: Temporarily stops program execution.
Line 110: Sets 8350A to remote and performs an Instrument Preset.
Line 120: Stops program execution.

To verify and investigate the different remote modes do the following:

1. Press CONTROL RESET SCRATCH END LINE on the 85A. This scratches the program memory.
2. Press INSTR PRESET on the 8350A.
3. Type in the above program.
4. Press SHIFT CLEAR RUN on the 85A.
5. With the 85A displaying "Remote", verify that the 8350A REMote light is lit. From the front panel, attempt to change the start frequency and verify that this is impossible. Verify that the Instrument Preset key and all other keys except LCL are disabled. Now press the LCL key and verify that the 8350A REMote light is off and that you can modify any of the sweep functions.
6. Press CONT on the 85A. With the 85A displaying "Local Lockout" verify that the 8350A REMote light is again lit. Again attempt to change the start frequency and perform an instrument preset. Verify that this is impossible. Now press the 8350A LCL key and verify that still no action is taken.
7. Press CONT on the 85A. With the 85A displaying "Local" verify that the 8350A REMote light is off. Also verify that all sweep functions now can be modified via the front panel controls.
8. Press CONT on the 85A. Verify that the 8350A has undergone an Instrument Preset and the REMote light is on. The Output ('OUTPUT 719') statement does two things, one it performs a 'REMOTE 719', and second it passes data to the 8350A.

Note that the 8350A LCL key produces the same result as programming 'LOCAL 719' or 'LOCAL 7'. Be careful as the latter command places all instruments on the HP-IB in local state as opposed to the 8350A alone.

EXAMPLE PROGRAM 2: Programming Functions

To program any function on the 8350A the controller must pass specific program codes and data to the sweeper. The statement that allows this is the Output ('OUTPUT') statement. The alphanumeric data string of the output statement can be a concatenation of character strings and/or variables. The data can be specific codes, free field formatted data, or reference a specific image ('IMAGE') statement. For example, to program the CW Frequency (CW), one program code sequence is "CW", followed by the frequency in GHz, then "GZ". If the frequency is to be 7.555 GHz, then the string "CW7.555GZ" will suffice. However if the frequency were to change then a variable 'F' could indicate the frequency in GHz and the program string could be "CW",F,"GZ". Using an image statement also allows a specific number of digits to be passed, thereby avoiding any unexpected round off errors.

NOTE

This program expects an 83500 Series Plug-in that covers the frequency 7.555 GHz. If using a plug-in that does not cover this frequency then the value in lines 20 and 30 should be changed to an appropriate value.

PROGRAM 2

```
 10 OUTPUT 719 ;"IP"
 20 OUTPUT 719 ;"CW7.555GZ"
 30 DISP "CW=7.555 GHz"
 40 PAUSE
 50 DISP "CW=(in GHz)=";
 60 INPUT F
 70 PRINT "CW=";F;"GHz"
 80 OUTPUT 719 ;"CW";F;"GZ"
 90 GOTO 50
100 IMAGE "CW",DD.DDD,"GZ"
110 OUTPUT 719 USING 100 ; F
120 GOTO 50
```

PROGRAM 2 EXPLANATION:

Line 10: Puts the 8350A into a predefined state via instrument preset.
Line 20: Puts the 8350A in CW mode and programs a CW frequency of 7.555 GHz.
Line 30: The 85A displays "CW = 7.555 GHz".
Line 40: Temporarily stops program execution.
Line 50: The 85A displays "CW (in GHz) =?".
Line 60: The user is prompted to input a new CW frequency value which is stored in the variable 'F'.
Line 70: Print on the CRT display the programmed CW frequency.
Line 80: Program the CW frequency using the default data format.
Line 90: Go to line 50.
Line 100: Image statement is set up for programming the CW frequency with a 1 MHz resolution.
Line 110: Program the CW frequency via image statement in line 100.
Line 120: Go to line 50.

The equipment setup is the same as the previous example. Reset the 85A, scratch the 85A memory, then type in the above program. Then perform the following:

1. Clear the 85A CRT display then run the program. The 85A displays "CW = 7.555 GHz". The 8350A changes from the instrument preset state of Start/Stop sweep to a CW frequency of 7.555 GHz.

2. Press CONT on the 85A. The 85A now displays "CW (in GHz) =?". Type in a new CW frequency (value in GHz), then press END LINE

3. The 8350A will be programmed to the new CW frequency with the new value printed on the internal printer. The program jumps back to step (2) above.

When inputting the CW frequency try several values, each with a different number of digits after the decimal point. Notice that the 8350A displays the frequency to 3 decimal places (1 MHz frequency resolution). Values with better than 1 MHz frequency resolution are rounded to the nearest MHz by the 8350A. However the 85A outputs data in a free-field format that outputs a number with all appropriate significant digits. Another approach is to utilize the image statement to set the desired number of decimal places. To use the image statement in the program, perform the following on the 85A:

Press PAUSE SHIFT CLEAR
Type 'DELETE 80, 90'
Press END LINE

This should delete lines 80 and 90 from program #2 and allow the use of lines 100, 110, and 120 instead. Run the modified program again and use the same steps for operation as before. Now if the value inputted has a frequency resolution greater than 1 MHz the 85A does the rounding instead of the 8350A. This is the preferred programming approach. Change the image statement for 10 MHz frequency

resolution and verify the results from the 8350A frequency display.

Since a device select code address can be a variable, verify that this can be used in the modified or original program #2 by doing the following:

1. Insert before line 10 a new line with the variable 'S' by:

   Press PAUSE SHIFT CLEAR
   Type '5 S=719'
   Press END LINE

2. Modify the output statement(s) by editing the necessary lines and changing the 'OUTPUT 719' to 'OUTPUT S' and 'OUTPUT 719 USING 100' to 'OUTPUT S USING 100'.

3. Re-run the modified program using the same operation steps as above.

EXAMPLE PROGRAM 3: Setting Up A Typical Sweep

Typically the sweeper is programmed for the proper sweep frequency range, sweep time, power level, and marker frequencies for a test measurement. This program sets up the sweeper for a general purpose situation using several dedicated image statements. Note that not all parameters need to be reprogrammed every time.

NOTE

This program expects an 83500 Series Plug-in that covers the frequency range of at least 3 to 7 GHz. If using a plug-in with a different frequency range, change the values in lines 60, 90, and 100, to the appropriate values. If using an 86200 Series Plug-in then do not enter line 70.

PROGRAM 3

```
 10 IMAGE "FA",DD.DDD,"GZFB",DD.
    DDD,"GZ"
 20 IMAGE "ST",DDDDD,"MS"
 30 IMAGE "M",D,DD.DDD,"GZ"
 40 IMAGE "PL",DDD.DD,"DM"
 50 OUTPUT 719 "IPMD1"
 60 OUTPUT 719 USING 10 ; 3,7
 70 OUTPUT 719 USING 40 ; 10
 80 OUTPUT 719 USING 20 ; 50
 90 OUTPUT 719 USING 30 ; 1,4
100 OUTPUT 719 USING 30 ; 2,6
110 END
```

PROGRAM 3 EXPLANATION:

Line 10: Image statement for setting the Start and Stop Sweep frequencies in GHz.
Line 20: Image statement for setting the Sweep Time in milliseconds.
Line 30: Image statement for setting a Frequency Marker by marker number and frequency in GHz.
Line 40: Image statement for setting the Output Power Level in dBm.
Line 50: Preset the sweeper to a known state via instrument preset and enable the internal 27.8 kHz Square Wave Amplitude Modulation.
Line 60: Set a Start/Stop Sweep of 3.0 to 7.0 GHz.
Line 70: Set the Output Power Level to +10 dBm.
Line 80: Set the Sweep Time to 50 milliseconds.
Line 90: Set Marker #1 to 4 GHz.
Line 100: Set Marker #2 to 6 GHz.
Line 110: Stop program execution.

Setup the equipment as shown in figure 2 by adding the 8755C, the 180TR or 182T, the 11664, and a test device like a 4 to 6 GHz Bandpass Filter. It is important that the two rear panel connections from the 8350A to the 8755C/182T are made for a proper CRT display. For the example measurement set the following front panel controls:

On the 8755C:
  Channel 1:
    Display ..... OFF (press all the display push buttons so that they are all out)
  Channel 2:
    Display ..... B
    dB/DIV ..... 10 dB
    Reference Level ..... −10 dB
    Reference Level Vernier ..... OFF
On the 182T or 180TR:
  Magnifier X1
  Display ..... INT

After connecting the equipment: reset the 85A, scratch the 85A memory, then type in the above program. Clear the 85A CRT display then run the program. The 8350A will initially undergo an instrument preset which will set the proper power leveling mode and sweep blanking signals. Since the 8755C requires the RF signal to be modulated at a 27.8 kHz rate, the internal amplitude modulation is



FIGURE 2: Equipment Setup For Program #3

enabled. If using a 4 to 6 GHz Bandpass Filter as the test device, the CRT display should reflect the filter transmission response over the 3 to 7 GHz range. Two frequency markers of the Z-Axis Intensity dot variety are set to 4 and 6 GHz, hopefully within the passband or near the 3 dB points. The setup can be modified by changing the values in lines 60, 70, 80, 90, and/or 100, then re-run the program.

EXAMPLE PROGRAM 4: Learning An Instrument State

Being able to save a specific instrument state is helpful when it is needed several times in a test or measurement procedure. The user could save the instrument state by manually logging the important sweep parameters such as frequency range, power level, ALC modes, etc., then re-inputting them at the appropriate time. A somewhat simpler approach is to save the instrument state in one of the 8350A internal storage registers, then recall it when needed. However, this is not a permanent solution unless the 8350A Non-volatile Memory option (Option 001) is used. A more permanent solution is to use the Output Learn String function of the 8350A so that the 85A can learn then store a data string that describes the present instrument state on a tape cartridge or in its' internal memory. Once an instrument state is stored or learned, the 8350A can then be restored to that state using the Input Learn String function. The power of these instrument Learn/Teach functions are demonstrated by the following program using the 85A fast data transfer function.

PROGRAM 4

```
 10 OPTION BASE 1
 20 DIM A$[100]
 30 IOBUFFER A$
 40 OUTPUT 719 ;"IPMD1"
 50 LOCAL 719
 60 PAUSE
 70 OUTPUT 719 ;"OL"
 80 TRANSFER 719 TO A$ FHS ; EOI
 90 A$=A$[1,90]
100 PAUSE
110 OUTPUT 719 ;"IL"&A$
120 END
```

PROGRAM 4 EXPLANATION:

Line 10: Define the first element of any array to be at index number 1.

Line 20: Set the length of the A$ string to 100 characters.

Line 30: Set up the string A$ as an I/O Buffer for data storage in fast read/write data transfer operations.

Line 40: Set the 8350A to a predefined state via instrument preset and enable the square wave modulation.

Line 50: Return the 8350A to local control.

Line 60: Temporarily stop program execution.

Line 70: Program the 8350A to output the Learn String.

Line 80: Read the Learn String into the 85A via the fast data transfer function using the HP-IB EOI (End or Identify) signal to terminate the transfer. Store the Learn String in A$.
Line 90: Extract the Learn String information from the I/O Buffer by removing the buffer pointers. Re-save the Learn String only in A$.
Line 100: Temporarily stop program execution.
Line 110: Program the 8350A to accept a Learn String, then send the new Learn String to the 8350A.
Line 120: Stop program execution.

Setup the equipment as in example 3 using the analyzers' CRT display to verify the sweep settings. Note that the original equipment setup can also be used with the 8350A front panel indicators used for verification. Reset the 85A, scratch the 85A memory, then type in the above program. Clear the 85A CRT display then run the program. The 8350A will undergo an instrument preset, enable the square wave modulation, then return to local front panel control. Then perform the following:

1. Adjust the 8350A to a preferred instrument state, then press CONT on the 85A.

2. Turn the 8350A line power off. Wait five seconds then turn the 8350A power back on. Press INSTR PRESET on the 8350A.

3. Press CONT on the 85A. Verify on the analyzers' CRT display and/or the 8350A that the original instrument state has been restored.

EXAMPLE PROGRAM 5: Interrogating The Present Value Of A Function

While the 8350A Learn String enables the user to completely save a string of characters that define the present instrument state, the information is densely packed and encoded to save memory space. If the user wishes to determine the actual value of a specific parameter, say the Start Frequency, it would require a tedious process to extract a numeric value from several characters within the Learn String. An easier approach is to use the Output Interrogated Parameter function of the 8350A. With this function the 85A instructs the 8350A to output the present numeric value of a specified function. Any function that has a numeric value associated with it (except Step Size) can be interrogated. Note that if the parameter is not presently active, the 8350A uses a computed value or its previous value. The following program demonstrates the capability of the interrogate function.

PROGRAM 5

```
10 OUTPUT 719 ;"IPMD1"
20 LOCAL 719
30 PAUSE
40 OUTPUT 719 ;"OPFA"
50 ENTER 719 ; A
60 PRINT "Start Freq=";A/100000
   0;"MHz"
70 OUTPUT 719 ;"OPFB"
80 ENTER 719 ; B
90 PRINT "Stop Freq=";B/1000000
   ;"MHz"
100 OUTPUT 719 ;"OPST"
110 ENTER 719 ; T
120 PRINT "Sweep Time=";1000*T;"
    msec"
130 END
```

PROGRAM 5 EXPLANATION:

Line 10: Set the 8350A to a predefined instrument state via instrument preset and enable the square wave modulation.
Line 20: Return the 8350A to local control.
Line 30: Temporarily stops program execution.
Line 40: Program the 8350A to output the present value of the Start Frequency.
Line 50: Read the value into the 85A and store it in the variable 'A'.
Line 60: Print on the internal printer the present value of the Start Frequency in MHz.
Line 70: Program the 8350A to output the present value of the Stop Frequency.
Line 80: Read the value into the 85A and store it in the variable 'B'.
Line 90: Print on the internal printer the present value of the Stop Frequency in MHz.
Line 100: Program the 8350A to output the present value of the Sweep Time.
Line 110: Read the value into the 85A and store it in the variable 'T'.
Line 120: Print on the internal printer the present value of the Sweep Time in milliseconds.
Line 130: Stops program execution.

Setup the equipment as in example 3 using the analyzers' CRT display to verify the sweep settings. Note that the original equipment setup can also be used with the 8350A front panel indicators used for verification. Reset the 85A, scratch the 85A memory, then type in the above program. Clear the 85A CRT display then run the program. The 8350A will

undergo an instrument preset, enable the square wave modulation, then return to local front panel control. Then perform the following:

1. Adjust the 8350A to a preferred instrument state using the Start Frequency, Stop Frequency, and Sweep Time controls.
2. Press CONT on the 85A.
3. The present values of the Start Frequency, Stop Frequency, and Sweep Time are sequentially interrogated and then printed on the internal printer of the 85A.

EXAMPLE PROGRAM 6: A Stepped CW Sweep

Present automatic measurement systems typically make measurements at a sequence of CW test frequencies instead of analog sweeping the frequency range of interest. If swept, the measurement data taking machine would need to sample the RF signal at a very fast rate to maintain accurate frequency information, too. This is typically not accomplished. Stepped CW sweeps can be accomplished in several ways with the 8350A:

1. Program sequential CW test frequencies.
2. Program the frequency sweep range then enable the manual sweep mode. Perform a stepped manual sweep by repetitively programming the step up/increment function.
3. Program the CW frequency to the start frequency, the Step Size to an appropriate value, then repetitively program the step up/increment function.

Considering the speed of programming the above approaches, the third is the most efficient time wise. This program illustrates a stepped sweep using this approach.

PROGRAM 6

```
10 OUTPUT 719 ;"IPMD1"
20 DISP "Start Freq (GHz)="
30 INPUT A
40 DISP "Stop Freq (GHz)=";
50 INPUT B
60 DISP "Step Size (GHz)=";
70 INPUT C
80 D=(B-A)/C
90 OUTPUT 719 ;"CWSS";C;"GZ"
100 OUTPUT 719 ;"CW";A;"GZ"
110 FOR I=1 TO D
120 OUTPUT 719 ;"UP"
130 WAIT 20
140 NEXT I
150 GOTO 100
```

PROGRAM 6 EXPLANATION:

Line 10: Set the 8350A to a predefined instrument state and enable the square wave modulation.
Line 20: The 85A displays "Start Freq (GHz) = ?".
Line 30: Input prompts for start frequency of the sweep. Store it in the variable 'A'.
Line 40: The 85A displays "Stop Freq (GHz) = ?".
Line 50: Input prompts for the stop frequency of the sweep. Store it in 'B'.
Line 60: The 85A displays "Step Size (GHz) = ?".
Line 70: Input prompts for the step size of the sweep. Store it in 'C'.
Line 80: Determine the number of frequency steps in sweep, store in 'D'.
Line 90: Set the CW Step Size.
Line 100: Set the CW frequency to the start frequency value.
Line 110: Iterate the CW step 'D' times.
Line 120: Program the Step Increment/Up function.
Line 130: Wait 20 milliseconds for settling.
Line 140: Continue step iteration.
Line 150: Go to line 100.

The equipment setup is the same as in the previous example. Reset the 85A, scratch the 85A memory, then type in the above program. Clear the 85A CRT display then run the program. The 8350A will undergo an instrument preset and enable the square wave modulation. Then perform the following:

1. The 85A will display "Start Freq (GHz) = ?". Answer this prompt by inputting the desired Start frequency (value in GHz) of the sweep, then press END LINE
2. The 85A will display "Stop Freq (GHz) = ?". Answer this prompt with the desired Stop frequency (in GHz) of the sweep, then press END LINE
3. The 85A will display "Step Size (GHz) = ?". Answer this prompt with the desired Step size (in GHz) of the sweep, then press END LINE
4. The 8350A CW frequency will be programmed to the Start frequency of the sweep selected. Then the CW frequency is repetitively incremented by the step size value. The sweep is then restarted after reaching the stop frequency.

To stop the program press STOP .

Since part of the time involved in changing CW frequencies is in updating the numeric LED display if this could be defeated the CW frequency time can be optimized. Note that one drawback is that the numeric display will not indicate the present frequency. The 8350A provides a Display Update On/Off function and it can be implemented by modifying line 10 to be:

OUTPUT 719 ;"IPMD1DU0"

Then re-run the modified program using the same operation steps as above.

EXAMPLE PROGRAM 7: Using Service Requests, Status Bytes, and Request Mask

Certain error conditions of the 8350A can be detected by the 85A so that corrective action can be taken. Examples of some detectable error conditions are RF power unleveled, numeric data entry out of range, and line power failure. If an error condition exists, the user can instruct the 8350A to request service from the 85A by initiating a Service Request (SRQ). The 85A can detect whether an SRQ has taken place on the bus by analyzing bit 7 (see note below) of the Status Byte of the 82937A HP-IB Interface. Two modes are available for analyzing the 82937A Status Byte: 1) periodically read the Status Byte, or 2) enable bit 7 to interrupt the program when it is set. In either case, once it is determined that the 8350A has requested service, the specific error condition(s) can then be determined by reading and analyzing the Status Bytes of the 8350A. The 8350A has two Status Bytes, each consisting of 8 bits with each bit indicating the present status of a particular function or condition. See Figure 3 for a complete description of the conditions associated with each Status Byte bit. The user can analyze these Status Bytes for every SRQ, or more simply, instruct the 8350A to issue an SRQ only if a specific set of error conditions exists. The set of conditions is determined by a numeric value passed by the Request Mask function. This numeric value is generated by summing the decimal values of each Status Byte bit to be checked. This program demonstrates the capability of the SRQ and Status Bytes to detect an error condition.

TABLE 1: 8350A Status Byte Descriptions

STATUS BYTE (#1)								
BIT #	**7**	**6**	**5**	**4**	**3**	**2**	**1**	**0**
DECIMAL VALUE	**128**	**64**	**32**	**16**	**8**	**4**	**2**	**1**
FUNCTION	Front Panel SRQ REQUEST	REQUEST SERVICE (RQS)	SRQ on Syntax Error	SRQ on End of Sweep	*SRQ on RF Settled	SRQ on Change in Extended Status Byte	SRQ on Front Panel Entry Complete	SRQ on Numeric Parameter Altered to Default Value

EXTENDED STATUS BYTE (#2)								
BIT #	**7**	**6**	**5**	**4**	**3**	**2**	**1**	**0**
DECIMAL VALUE	**128**	**64**	**32**	**16**	**8**	**4**	**2**	**1**
FUNCTION	N/A	*RF Unleveled	Power Failure	*RF Unlocked	*External Freq. Ref. Selected	*OvenCold	*Over Modulation	Self Test Failed

*Bit/Functions not usable with 86200 Series Plug-ins and 11869A Adapter.

NOTE

This assumes that the status bits are numbered 0 thru 7 with the least-significant bit being number 0. Other references may assume that the bits are numbered 1 thru 8 with the least-significant bit being number 1.

If using an 86200 Series Plug-in, the Status Bytes can provide only limited information. Table 1 indicates which Status Byte functions/bits are usable.

PROGRAM 7

```
 10 ABORTIO 7
 20 CLEAR 719
 30 OUTPUT 719 ;"IPMD1"
 40 ON INTR 7 GOSUB 110
 50 ENABLE INTR 7;8
 60 DISP "CW Freq (GHz)=";
 70 INPUT F
 80 OUTPUT 719 ;"CW";F;"GZ"
 90 WAIT 100
100 GOTO 60
110 STATUS 7,1 ; X
120 A=SPOLL(719)
130 IF BIT(A,0)=1 THEN PRINT "Pa
    rameter Altered"
140 IF BIT(A,5)=1 THEN PRINT "Sy
    ntax Error"
150 CLEAR 719
160 ENABLE INTR 7;8
170 RETURN
```

PROGRAM 7 EXPLANATION:

Line 10: Clear the status of the HP-IB.
Line 20: Clear the status of the 8350A.
Line 30: Preset the 8350A to a predefined instrument state and enable the square wave modulation.
Line 40: Indicate that if an interrupt from the 82937A HP-IB Interface is received that program execution will branch to the interrupt service routine located at the line 110.
Line 50: Specify and enable the controller to accept an interrupt from the 82937A if bit 3 (decimal value 8) is set.
Line 60: The 85A displays "CW Freq (GHz) =?".
Line 70: Input prompts for the desired CW frequency value in GHz. Store it in the variable 'F'.
Line 80: Set the CW frequency as determined by 'F'.
Line 90: Wait 100 milliseconds to allow the 8350A to interrupt.
Line 100: Go to line 60.
Line 110: Read the 82937A interrupt cause register to enable another interrupt.
Line 120: Location of the interrupt service routine. Read the Status Byte of the 8350A and store it in 'A'.
Line 130: Check bit 0 of the 8350A Status Byte for an Altered Parameter error. Print on the internal printer "Parameter Altered" if one exists.
Line 140: Check bit 5 of the 8350A Status Byte for a Syntax Error. Print on the internal printer "Syntax Error" if one exists.
Line 150: Clear the status of the 8350A.
Line 160: Re-specify and re-enable bit 3 of the 82937A to cause an interrupt.
Line 170: Return from the interrupt service routine to the main program.

The equipment setup is the same as the previous example. Reset the 85A, scratch the 85A memory, then type in the above program. Clear the 85A CRT display then run the program. The 8350A will undergo an instrument preset and enable the square wave modulation. The 85A then displays "CW Freq (GHz) = ?". Answer this prompt by inputting the desired CW frequency in GHz, then press END LINE. Verify that the 8350A CW frequency has been properly programmed. Try several values that are out of range of the plug-in's frequency limits and verify that an error message was printed on the internal printer. The program repeats the above input prompt. To stop the program press PAUSE

† NOTE

For Program 7 to function properly change line 30 to: 30 OUTPUT 719; "IPMD1RM & CHR$(97). This change enables bit 5 (SRQ on Syntax Error) and bit 0 (SRQ on Numeric Parameter to Default Value).

HP-IB PROGRAM CODES

CODE	DESCRIPTION
AKm	Amplitude Marker On/Off
ALmn	Alternate Sweep On/Off
A1	Internal Leveling
A2	External Crystal Leveling
A3	External Power Meter Leveling
BK	Backspace
CAm	Amplitude Crystal Marker On/Off (83522/83525 Only)
CF	Center Frequency
CIm	Intensity Crystal Marker On/Off (83522/83525 Only)
CW	CW Frequency
C1	1 MHz Crystal Marker Frequency (83522/83525 Only)
C2	10 MHz Crystal Marker Frequency (83522/83525 Only)
C3	50 MHz Crystal Marker Frequency (83522/83525 Only)
C4	External Crystal Marker Frequency (83522/83525 Only)
DF	Delta F Frequency Span
DM	dBm
DN	Step Down/Decrement
DPm	Display Blanking On/Off
DUm	Display Update On/Off
E	Exponent Power Of 10
FA	Start Frequency
FB	Stop Frequency
FIm	CW Filter In/Out
GZ	GHz
HZ	Hz
IL	Input Learn String
IP	Instrument Preset
IX	Input Micro Learn String
KZ	KHz
MC	Marker To Center Frequency
MDm	Square Wave Amplitude Modulation On/Off
MO	Marker Off
MPm	Marker 1-2 Sweep On/Off
MS	Milliseconds
MZ	MHz
M0	Marker Off
M1	Marker #1
M2	Marker #2
M3	Marker #3
M4	Marker #4
M5	Marker #5
NT	Network Analyzer Trigger (8410B)
OA	Output Active Parameter
OL	Output Learn String
OM	Output Mode String
OP	Output Interrogated Parameter
OS	Output Status Bytes
OX	Output Micro Learn String
PL	Power Level
PSm	Power Sweep On/Off
RCn	Recall Register
RFm	RF Power On/Off
RM	Service Request Mask
RS	Reset Sweep
SC	Seconds
SH	Shift Function
SLm	Slope On/Off
SM	Manual Sweep
SS	Step Size
ST	Sweep Time
SVn	Save Register
SX	External Sweep
TS	Take Sweep
T1	Internal Sweep Trigger
T2	Line Sweep Trigger
T3	External Sweep Trigger
T4	Single Sweep
UP	Step Up/Increment
VR	CW Vernier
0—9 + —	Acceptable Numeric Data

NOTES

1. **Program codes of the form "XXm" use "m" to turn the function On or Off (1 or 0). For the storage register functions the "n" is 1 through 9.**
2. **The 8350A ignores spaces, plus signs, negative signs (except for vernier, offset, and power values), and any unexpected characters. Program codes can be upper or lower case alpha characters.**

For more information, call your local HP Sales Office or nearest Regional Office: **Eastern** (201) 265-5000; **Midwestern** (312) 255-9800; **Southern** (404) 955-1500; **Western** (213) 970-7500; **Canadian** (416) 678-9430. Ask the operator for instrument sales. Or write Hewlett-Packard, 1501 Page Mill Road, Palo Alto, CA 94304. **In Europe:** Hewlett-Packard S.A., 7, rue du Bois-du-Lan, P.O. Box, CH 1217 Meyrin 2, Geneva, Switzerland. **In Japan:** Yokogawa-Hewlett-Packard Ltd., 29-21, Takaido-Higashi 3-chome, Suginami-ku, Tokyo 168.

5952-9308

Printed in U.S.A.